

The LU Decomposition

Worksheet by Russell Blyth and Mike May, S.J.

```
> restart: with(LinearAlgebra): with(plots): with(plottools):  
Warning, the name changecoords has been redefined  
Warning, the assigned name arrow now has a global binding
```

Outline

This worksheet investigates the LU decomposition of a matrix.
The basic objectives are:

- 1) Compute the LU decomposition, if possible.
- 2) See how elementary matrices are involved in the process.
- 3) Understand how the LU decomposition is useful in solving systems of equations.

Note: If you need to re-execute any command in this worksheet you should re-execute the entire worksheet from the first command line above, since variable names are carried through the worksheet.

```
>
```

Finding an LU Decomposition

First define a random 3 x 4 matrix.

```
> A := RandomMatrix(3,4,generator=rand(-5..5));
```

$$A := \begin{bmatrix} 1 & 4 & 0 & -4 \\ 5 & -2 & 0 & -1 \\ 5 & -5 & 2 & -1 \end{bmatrix} \quad (2.1)$$

Let's keep track of the elementary matrices corresponding to the row operations used, as well as the cumulative product of the elementary matrices, which we can do by starting with the matrix A augmented by the appropriate identity matrix:

```
> Iden:=IdentityMatrix(3);  
AAug := <A | Iden>;
```

$$Iden := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$
$$AAug := \begin{bmatrix} 1 & 4 & 0 & -4 & 1 & 0 & 0 \\ 5 & -2 & 0 & -1 & 0 & 1 & 0 \\ 5 & -5 & 2 & -1 & 0 & 0 & 1 \end{bmatrix} \quad (2.2)$$

Now we apply elementary row operations to the augmented matrix to get a row echelon form. We intend to use only elementary row operations of the form that add multiples of rows to lower rows (in order to get an LU-decomposition). We exhibit the elementary matrices at each step. (Note that the matrix A is not truly "random" - every time the worksheet is started, it produces the same matrix A. Thus we know in advance which row operations are required in this worksheet. This is handy for teaching purposes. If you want a truly "random" matrix, you need to precede the randmatrix command with a "randomize()", which generates a seed for the random functions in Maple, based on the computer clock.)

```
> type(AAug, Matrix);
```

true (2.3)

```
> L1 := -AAug[2,1]/AAug[1,1];
AAugR := RowOperation(AAug, [2,1], L1);
E1 := RowOperation(Iden, [2,1], L1);
```

L1 := -5

$$AAugR := \begin{bmatrix} 1 & 4 & 0 & -4 & 1 & 0 & 0 \\ 0 & -22 & 0 & 19 & -5 & 1 & 0 \\ 5 & -5 & 2 & -1 & 0 & 0 & 1 \end{bmatrix}$$

$$E1 := \begin{bmatrix} 1 & 0 & 0 \\ -5 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

(2.4)

```
> L2 := -AAugR[3,1]/AAugR[1,1];
AAugR := RowOperation(AAugR, [3,1], L2);
E2 := RowOperation(Iden, [3,1], L2);
```

L2 := -5

$$AAugR := \begin{bmatrix} 1 & 4 & 0 & -4 & 1 & 0 & 0 \\ 0 & -22 & 0 & 19 & -5 & 1 & 0 \\ 0 & -25 & 2 & 19 & -5 & 0 & 1 \end{bmatrix}$$

$$E2 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ -5 & 0 & 1 \end{bmatrix}$$

(2.5)

```
> L3 := -AAugR[3,2]/AAugR[2,2];
AAugR := RowOperation(AAugR, [3,2], L3);
E3 := RowOperation(Iden, [3,2], L3);
```

L3 := -25/22

$$AAugR := \begin{bmatrix} 1 & 4 & 0 & -4 & 1 & 0 & 0 \\ 0 & -22 & 0 & 19 & -5 & 1 & 0 \\ 0 & 0 & 2 & -57/22 & 15/22 & -25/22 & 1 \end{bmatrix}$$

$$E3 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & \frac{-25}{22} & 1 \end{bmatrix} \quad (2.6)$$

AAugR is in row echelon form. Note we don't bother to force leading ones. The matrix U can now be extracted:

```
> U := DeleteColumn(AAugR, 5..7);
```

$$U := \begin{bmatrix} 1 & 4 & 0 & -4 \\ 0 & -22 & 0 & 19 \\ 0 & 0 & 2 & \frac{-57}{22} \end{bmatrix} \quad (2.7)$$

The matrix B referred to in class is the product of the three elementary matrices used. This matrix B can also be extracted from AAugR:

```
> E3.E2.E1;
B := DeleteColumn(AAugR, 1..4);
```

$$B := \begin{bmatrix} 1 & 0 & 0 \\ -5 & 1 & 0 \\ \frac{15}{22} & \frac{-25}{22} & 1 \end{bmatrix} \quad (2.8)$$

The matrix L we seek is just the inverse of this B:

```
> L := MatrixInverse(B);
```

$$L := \begin{bmatrix} 1 & 0 & 0 \\ 5 & 1 & 0 \\ 5 & \frac{25}{22} & 1 \end{bmatrix} \quad (2.9)$$

Now note the remarkable relationship between the entries of L, the elementary matrices, and the multipliers used in the row reductions! The matrix L can in fact be written down without actually computing the inverse of B.

The off diagonal entries of L are the negatives of the multipliers used in the row operations.

The off diagonal entries of L are also the off diagonal entries of the inverses of the elementary matrices corresponding to the row operations.

```
> [-L1, -L2, -L3];
[E1^(-1), E2^(-1), E3^(-1)];
```

$$\begin{bmatrix} 5, 5, \frac{25}{22} \end{bmatrix}$$

$$\left[\begin{bmatrix} 1 & 0 & 0 \\ 5 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 5 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & \frac{25}{22} & 1 \end{bmatrix} \right] \quad (2.10)$$

Check that $A = LU$

```
> (L . U); (A);
```

$$\begin{bmatrix} 1 & 4 & 0 & -4 \\ 5 & -2 & 0 & -1 \\ 5 & -5 & 2 & -1 \\ 1 & 4 & 0 & -4 \\ 5 & -2 & 0 & -1 \\ 5 & -5 & 2 & -1 \end{bmatrix}$$

(2.11)

Exercise

1) Repeat the above for a random 4 x 4 matrix AA (be careful not to reuse any of the variable names used previously, since some get reused below!)

```
>
```

```
>
```

Using the LU decomposition to solve a system

Now let's see how to use the LU decomposition to solve a system $AX = Y$.

We showed above that this is equivalent to a system $L(UX) = Y$. We can solve this system in two steps using only substitution on a triangular system in each step. Let $Z=UX$.

Generate a RHS vector Y.

```
> Y := RandomMatrix(3,1,generator=rand(-5..5));
```

$$Y := \begin{bmatrix} 4 \\ 5 \\ -4 \end{bmatrix}$$

(3.1)

Solve $LZ = Y$ for Z; note this amounts to a simple forward-substitution (no further row reductions).

We show the augmented matrix and have Maple compute.

```
> <L | Y>;
Z:=LinearSolve(L,Y);
```

$$\begin{bmatrix} 1 & 0 & 0 & 4 \\ 5 & 1 & 0 & 5 \\ 5 & \frac{25}{22} & 1 & -4 \end{bmatrix}$$

$$Z := \begin{bmatrix} 4 \\ -15 \\ -\frac{153}{22} \end{bmatrix} \quad (3.2)$$

Then solve $UX = Z$ for X ; note this amounts to a simple back-substitution (no further row reductions). We show the augmented matrix and again have Maple compute the solution for us.

```
> <U|Z>;
X := LinearSolve(U,Z);
```

$$\begin{bmatrix} 1 & 4 & 0 & -4 & 4 \\ 0 & -22 & 0 & 19 & -15 \\ 0 & 0 & 2 & \frac{-57}{22} & \frac{-153}{22} \end{bmatrix}$$

$$X := \begin{bmatrix} \frac{14}{11} + \frac{6}{11} -t_{0,1} \\ \frac{15}{22} + \frac{19}{22} -t_{0,1} \\ -\frac{153}{44} + \frac{57}{44} -t_{0,1} \\ -t_{0,1} \end{bmatrix} \quad (3.3)$$

(Maple does not always select the parameter the way we do.)

Compare the result of solving the original system.

```
> LinearSolve(A,Y);
```

$$\begin{bmatrix} \frac{14}{11} + \frac{6}{11} -t_{1,1} \\ \frac{15}{22} + \frac{19}{22} -t_{1,1} \\ -\frac{153}{44} + \frac{57}{44} -t_{1,1} \\ -t_{1,1} \end{bmatrix} \quad (3.4)$$

Exercises

- (2) Generate a random 4×1 vector YY , and then repeat the steps above to solve $(AA)(X) = YY$.

>

3) What advantage would the LU decomposition have for computer computation of solutions to linear systems?

>

4) Use Maple's help command to find out about the LUdecomp command in the linalg package, Use the command to check your LU decomposition of AA in Exercise 1.

>

>

>